Calcul de la somme d'une série entière

Pour $n \in \mathbf{N}$, on pose $a_n = \int_0^1 \left(\frac{1+t^2}{2}\right)^n \mathrm{d}t$.

1/ Vérifier la relation : $\forall n \in \mathbf{N}^*, \quad (2n+1)u_n = 1 + nu_{n-1}$.

2/ Montrer que la suite (a_n) est positive, décroissante et minorée par $\left(\frac{1}{2n+1}\right)$.

3/ Déduire, de ce qui précède, que la série entière $\sum_{n\geqslant 0} a_n x^n$ converge pour $x = -1$ et diverge pour $x = 1$. Préciser le rayon de convergence de la série.

4/ Calculer la somme de la série $\sum_{n\geqslant 0} a_n x^n$.

1 Avec *Mathematica* 5.2

```
<<DiscreteMath`RSolve`;
```

Obtention de la série entière dont les coefficients sont définis à l'aide d'une récurrence.

```
f[x_] = GeneratingFunction[{(2n+1)a[n]==1 + n a[n-1], a[0]==1}, a[n], n, x] [[1,1]]
```

$$-\frac{2\arctan\left(\frac{1}{\sqrt{\frac{x-2}{x}}}\right)}{\sqrt{x-2}\sqrt{x}}$$

Simplification de l'expression pour $-1 < x < 0$. Les termes sous les racines sont regroupés de façon appropriée.

```
FullSimplify[f[x], -1 < x < 0]
```

$$-\frac{2\arctan\left(\frac{x}{\sqrt{(x-2)x}}\right)}{\sqrt{(x-2)x}}$$

Simplification de l'expression pour $0 < x < 1$. Cela ne marche pas vraiment !

```
FullSimplify[f[x], 0 < x < 1]
```

$$-\frac{2\arctan\left(\frac{1}{\sqrt{\frac{x-2}{x}}}\right)}{\sqrt{(x-2)x}}$$

Nous allons forcer un peu les choses...

```
FullSimplify[Re[ComplexExpand[f[x]]], 0 < x < 1]
```

$$\frac{\log\left(x+\sqrt{-(x-2)x^3}\right)-\log\left(x-\sqrt{-(x-2)x^3}\right)}{2\sqrt[4]{(x-2)^2x^2}}$$

2 Avec *Mathematica* 6.03

2.1 La relation de récurrence

Définition de a_n en deux temps, en commençant par la fonction dont a_n est une intégrale.

```
f[n_, t_] := ((1 + t^2)/2)^n
```

```
a[n_] := Integrate[f[n, t], {t, 0, 1}]
```

Quelques valeurs pour voir.

```
Table[a[i], {i, 0, 5}]
```

$$\left\{1, \frac{2}{3}, \frac{7}{15}, \frac{12}{35}, \frac{83}{315}, \frac{146}{693}\right\}$$

L'expression de a_n.

```
a[n]
```

$$2^{-n}\,{}_2F_1\left(\frac{1}{2}, -n; \frac{3}{2}; -1\right)$$

On retrouve donc les **séries hypergéométriques**.

Testons la relation de récurrence :

```
(2n+1)a[n] - 1 - n a[n-1]
```

$$i2^{-n}nB_{-1}\left(\frac{1}{2}, n\right) + 2^{-n}(2n+1)\,{}_2F_1\left(\frac{1}{2}, -n; \frac{3}{2}; -1\right) - 1$$

Forçons les choses.

```
% // FullSimplify
```

$$0$$

La relation est vraie. Elle peut être vérifiée, aussi, en la reprenant au niveau des fonctions à intégrer.

```
g[n_, t_] := (2 n + 1) f[n, t] - 1 - n f[n - 1, t]
```

```
Integrate[g[n, t], {t, 0, 1}]
```

$$0$$

2.2 La somme

Nous permutons les signes $\sum$ et $\int$ en distinguant deux cas.

```
Integrate[Sum[((1+t^2)/2)^n x^n, {n,0,Infinity}], {t,0,1},
    Assumptions -> -1 < x < 0]
```

$$\frac{2\arctan\left(\sqrt{\frac{x}{x-2}}\right)}{\sqrt{(x-2)x}}$$

```
Integrate[Sum[((1+t^2)/2)^n x^n, {n,0,Infinity}], {t,0,1},
    Assumptions -> 0 < x < 1]
```

$$\frac{2\,\text{argcoth}\left(\sqrt{\frac{2}{x}-1}\right)}{\sqrt{-(x-2)x}}$$

3 Avec *Mathematica* 7

3.1 Utilisation de `GeneratingFunction`

`GeneratingFunction` est intégrée au système de base.

```
a[n] /. First[RSolve[{(2n+1)a[n]==1 + n a[n-1], a[0]==1}, a[n], n]]
```

$$-\frac{2^{-n}(1)_n\left(2^{n+1}\left(\frac{1}{2}(2n+1)\right)!\,(;n+2;2)+i\sqrt{\pi}(n+1)!\right)}{\sqrt{\pi}(n+1)!\left(\frac{3}{2}\right)_n}$$

Il semble y avoir un bug dans la transformation en LaTeX de l'écriture des fonctions hypergéométriques...

```
FunctionExpand[%]
```

$$-\frac{(;n+2;2)}{n+1}-\frac{i\sqrt{\pi}2^{-n-1}\Gamma(n+1)}{\Gamma\left(n+\frac{3}{2}\right)}$$

```
GeneratingFunction[%, n, x]
```

$$-\text{GeneratingFunction}\left[\frac{(;n+2;2)}{n+1},n,x\right]-\frac{2i\arcsin\left(\frac{\sqrt{x}}{\sqrt{2}}\right)}{\sqrt{-(x-2)x}}$$

La somme n'a donc pas été calculée complètement... Les deux étapes précédentes ne semblent pas reproduire ce qui était mis à disposition dans la version 5.2 !

Il faudrait explorer le paquet `RSolve.m` de la version 5.2...

3.2 La somme

Nous permutons les signes $\sum$ et $\int$ en distinguant deux cas.

```
Integrate[Sum[((1+t^2)/2)^n x^n, {n,0,Infinity}], {t,0,1},
    Assumptions -> -1 < x < 0]
```

$$\frac{2\arctan\left(\sqrt{\frac{x}{x-2}}\right)}{\sqrt{(x-2)x}}$$

```
Integrate[Sum[((1+t^2)/2)^n x^n, {n,0,Infinity}], {t,0,1},
    Assumptions -> 0 < x < 1]
```

$$\frac{2\,\operatorname{argcoth}\left(\sqrt{\frac{2}{x}-1}\right)}{\sqrt{-(x-2)x}}$$